Problem:
Score : 100 points

Problem Statement

You are given strings s and t.
Find one longest string that is a subsequence of both s and t.

Notes

A subsequence of a string x is the string obtained by removing zero or more characters from x and concatenating the remaining characters without changing the order.

Constraints

s and t are strings consisting of lowercase English letters.

1 \leq |s|, |t| \leq 3000

Input

Input is given from Standard Input in the following format:

s
t

Output

Print one longest string that is a subsequence of both s and t.
If there are multiple such strings, any of them will be accepted.

Sample Input 1

axyb
abyxb

Sample Output 1

axb

The answer is axb or ayb; either will be accepted.

Sample Input 2

aa
xayaz

Sample Output 2

aa

Sample Input 3

a
z

Sample Output 3

The answer is (an empty string).

Sample Input 4

abracadabra
avadakedavra

Sample Output 4

aaadara

Solution in Common Lisp (SBCL 2.0.3) (judge verdict: Accepted)

(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx opt
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(in-package :cl-user)

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun main ()
  (declare #.OPT)
  (let* ((ss (coerce (read-line) 'simple-base-string))
         (slen (length ss))
         (tt (coerce (read-line) 'simple-base-string))
         (tlen (length tt))
         (dp (make-array '(3001 3001) :element-type 'uint16 :initial-element 0))
         (prev-xs (make-array '(3001 3001) :element-type 'uint16 :initial-element 0))
         (prev-ys (make-array '(3001 3001) :element-type 'uint16 :initial-element 0))
         (cs (make-array '(3001 3001) :element-type 'base-char :initial-element #\Nul)))
    (dotimes (x (+ slen 1))
      (dotimes (y (+ tlen 1))
        (unless (= x y 0)
          (let ((value -1)
                (prev-x 0)
                (prev-y 0)
                (char #\Nul))
            (when (and (> x 0) (> (aref dp (- x 1) y) value))
              (setq value (aref dp (- x 1) y)
                    prev-x (- x 1)
                    prev-y y))
            (when (and (> y 0) (> (aref dp x (- y 1)) value))
              (setq value (aref dp x (- y 1))
                    prev-x x
                    prev-y (- y 1)))
            (when (and (> x 0) (> y 0)
                       (char= (aref ss (- x 1)) (aref tt (- y 1)))
                       (> (+ 1 (aref dp (- x 1) (- y 1))) value))
              (setq value (+ 1 (aref dp (- x 1) (- y 1)))
                    prev-x (- x 1)
                    prev-y (- y 1)
                    char (aref ss (- x 1))))
            (setf (aref dp x y) value
                  (aref prev-xs x y) prev-x
                  (aref prev-ys x y) prev-y
                  (aref cs x y) char)))))
    (let ((x slen)
          (y tlen)
          res)
      (loop until (= x y 0)
            unless (char= (aref cs x y) #\Nul)
            do (push (aref cs x y) res)
            do (psetq x (aref prev-xs x y)
                      y (aref prev-ys x y)))
      (map () #'write-char res)
      (terpri))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "axyb
abyxb
"
    "axb
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "aa
xayaz
"
    "aa
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "a
z
"
    "
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "abracadabra
avadakedavra
"
    "aaadara
")))